# Maintainer: Jonas Witschel <diabonas@archlinux.org>
pkgname=helvum
pkgver=0.3.4
pkgrel=1
pkgdesc='GTK patchbay for PipeWire'
arch=('loong64' 'x86_64')
url='https://gitlab.freedesktop.org/pipewire/helvum'
license=('GPL3')
depends=('gtk4' 'pipewire')
makedepends=('appstream-glib' 'cargo' 'clang' 'meson')
source=("$url/-/archive/$pkgver/helvum-$pkgver.tar.gz")
sha512sums=('f8c73855a52e39337081ea8cf1e813444862ce8c214df7870b01c0064ed130e98074aaf305e982f1a50cc7e166d662b22246c2577c68043feca89cc325c1050f')

build() {
	meson --prefix=/usr --buildtype=plain "$pkgname-$pkgver" build
	meson compile -C build
}

package() {
	meson install -C build --destdir "$pkgdir"
}
